Extraterrestrial Solar Radiation on Inclined Surfaces

C. David Whiteman and K. Jerry Allwine

Pacific Northwest Laboratory, Richland, Washington 99352, USA

Abstract

A set of equations and a FORTRAN program have been developed to calculate extraterrestrial solar radiation on a plane surface of arbitrary inclination and azimuth at any point on the Earth. The model inputs are the day of year, local standard time, longitude, latitude, and azimuth and inclination angles of the inclined plane. The outputs are instantaneous solar flux, and solar zenith and azimuth angles. Daily integrated solar flux on the plane surface can also be calculated, as well as the local standard time of sunrise and sunset. Examples are shown to illustrate the use of the program.

Key Words: extraterrestrial solar radiation, radiation on slopes, solar geometry, equation of time

1. INTRODUCTION

Radiative flux from the sun provides the main source of energy to the Earth, providing the energy that is utilized in life processes in the Earth's ecosystems and that drives the surface energy balance, the water cycle, and atmospheric and oceanic circulations. Calculations of solar radiation input to the Earth are a function of the changing spherical trigonometric relationships between the sun and the Earth as the Earth revolves in orbit about the sun. Calculations of solar flux are essential in a number of disciplines, such as forestry, meteorology, hydrology, architecture, and astronomy. Formulas for such calculations have been available for many years (e.g., [1]). In this paper we present equations derived by previous investigators [1,2,3] to calculate extraterrestrial solar flux, and combine several of these formulas to develop a set of simple equations and a FORTRAN program to calculate extraterrestrial solar flux on inclined planes of arbitrary inclination and azimuth angle at any point on the Earth's surface. The extraterrestrial solar flux calculations specify the flux at the "edge" of the Earth's atmosphere. Those who wish to calculate solar flux at the Earth's surface may wish to develop an expanded version of the model accounting for depletion of the solar beam by absorption, reflection and refraction as it traverses the Earth's atmosphere. A number of previous investigators have developed equations that can be used for this purpose.

2. EQUATIONS

2.1 Trigonometric Equations

Sellers [1] provides a useful reference for the trigonometric formulas used to calculate extraterrestrial solar flux on an inclined surface. He provides an equation (Eqn. 3.15, p. 35) that requires the calculation of the solar zenith and azimuth angles for each determination of instantaneous solar flux. Radiation on the slope is seen to be a function of the solar constant, its modulation due to the changing distance between the sun and the Earth as the Earth revolves around the sun, and the angle between the unit vector normal to the slope and the solar beam. A somewhat more convenient formula was developed by Garnier and Ohmura [2], who used a coordinate transformation to derive a single, simplified equation in which the angle between the vector pointing from the surface to the sun and the unit normal vector pointing out of the surface can be determined by specifying the latitude of the surface, the azimuth and zenith angles of the inclined plane, and the solar declination.

This formula is an improvement over previous formulas since it does not require calculation of the sun's azimuth and zenith angles. Using Garnier and Ohmura's formula we can write the basic equation for solar radiation on an inclined plane as:

$$Q_{Si} = S_o \left(\frac{\bar{d}}{d}\right)^2 \cos\beta = S_o \left(\frac{\bar{d}}{d}\right)^2 [(\sin\phi \cos h)(-\cos a \sin i)$$

$$- \sin h (\sin a \sin i) + (\cos\phi \cos h) \cos i] \cos\delta$$

$$+ [\cos\phi (\cos a \sin i) + \sin\phi \cos i] \sin\delta \qquad (1)$$

where $\bar{d}$ = mean Earth-sun distance,
d = actual Earth-sun distance,
β = angle between unit normal vector of slope and vector to sun,
ϕ = latitude (+ for northern Hemisphere, - for southern hemisphere),
h = hour angle from solar noon (negative before noon, 15 deg/hour),
a = slope azimuth angle (true north),
i = slope inclination angle (from horizontal),
δ = sun's declination ($-23^\circ 26' < \delta < 23^\circ 26'$; positive when the sun is north of the equator), and
S_o = solar constant (1353 W/m^2).

McCullough [3] has given a convenient formula for the correction of the solar constant necessitated by the changing distance between the sun and Earth as the Earth travels in its annual orbit about the sun. The formula is as follows:

0266-9838/86/020164 - 06 $2.00

Paper received on April 28, 1986, and in final form on August 2, 1986. Referee: Prof. Robert D. Bornstein

$$(\frac{\bar{d}}{d})^2 = (1 - e \cos \omega D)^{-2} \quad (2)$$

where e = eccentricity of the Earth's orbit (0.0167),
ω = angular velocity of Earth in its orbit about the sun (360/365 degrees per day), and
D = Julian day (Jan. 1 = 1, Dec. 31 = 365).

McCullough [3] also provides a simple formula for the solar declination:

$$\delta = \sin^{-1} (\sin \varepsilon \sin \xi), \quad (3)$$

where ε = maximum solar declination (23° 26′), and
ξ = longitude of Earth in its orbit about the sun, as reckoned from the Julian date of the Vernal equinox, D_o = 80.

The Earth's orbital longitude is, in turn, given by the formula:

$$\xi = \omega(D - D_o) + 2 e (\sin \omega D - \sin \omega D_o) \quad (4)$$

All of the equations above can be evaluated without specific calculation of the sun's position angles, i.e., the solar zenith and azimuth angles. For reference, these angles can be calculated [1] with the formulas:

$$Z = \cos^{-1} (\sin \phi \sin \delta + \cos \phi \cos \delta \cos h) \quad (5)$$

and

$$A = 180^0 - \cos^{-1} \frac{\sin \phi \cos Z - \sin \delta}{\cos \phi \sin Z} \quad (6)$$

where the azimuth of the sun, A, is measured relative to true north.

The hour angles of sunrise and sunset [1] on a horizontal surface at the latitude of the site in question are given by:

$$h_{SR} = -H, \text{ and}$$
$$h_{SS} = H, \quad (7)$$

$$\text{where } H = \cos^{-1} (-\tan \phi \tan \delta). \quad (8)$$

Sunrise on an inclined plane can be delayed relative to the time of sunrise on a horizontal plane if the plane is tilted away from the sun at sunrise. Similarly, sunset can be delayed if the inclined plane tilts away from the late afternoon sun. This is expressed by the equations:

$$h_{SR} = \max (-H, h_{SRi}), \text{ and}$$
$$h_{SS} = \min (H, h_{SSi}), \quad (9)$$

where h_{SRi} and h_{SSi} can be determined by setting Eqn. (1) to zero and solving for the hour angle using an iterative convergence technique. This procedure for calculating the sunrise and sunset times on the inclined plane is avoided in the present program by simply calculating the course of instantaneous radiation on the inclined plane beginning at -H and proceeding to H at hour angle intervals equivalent to one minute. The sunrise hour angle on the inclined plane occurs at -H or at the hour angle when the instantaneous radiation on the inclined plane changes from negative to positive values, whichever value is larger. Similarly, the sunset hour angle occurs at H or at the hour angle when the instantaneous radiation changes from positive to negative, whichever value is smaller.

Daily total solar radiation can be obtained from the instantaneous values calculated with the above formulas by integration of the instantaneous values from sunrise to sunset. Daily totals are reported in MJ/m^2, in keeping with the Systeme International (SI) units used elsewhere in the text. Minor changes in the FORTRAN program are required to obtain instantaneous or daily total calculations in other units. For those who wish to perform calculations in English units, an equivalent solar constant value of 429 $Btu/h/ft^2$ is suggested [5].

Special considerations are necessary to calculate instantaneous and daily total extraterrestrial radiation at latitudes within the Arctic and Antarctic Circles, where

$$|\phi| \geq 90^0 - |\delta|. \quad (10)$$

Polar daylight (i.e., 24-hour-long daylight) will occur for latitudes in the Northern Hemisphere, satisfying Eqn. (10) when δ is positive, and in the Southern Hemisphere when δ is negative. Polar night will occur at identical latitudes of the opposite hemisphere. Instantaneous and daily total radiation values during polar night are zero. Instantaneous radiation values for the polar day can be calculated for any time during the 24-hour period, and daily totals are calculated for the 24-hour period beginning at midnight.

2.2 Equations for Calculating Time

In the above equations, times are given in terms of hour angle, and thus are referenced to solar noon. The Local Standard Time (LST) corresponding to a given hour angle in degrees can be computed from the equation:

$$t = 12 + \frac{h}{15} + C + \frac{4}{60} (\lambda - \lambda_o) \quad \text{[hours, LST]} \quad (11)$$

The first two terms on the right side of this equation compute the local solar time that corresponds to hour angle h, where a 15-degree hour angle corresponds to one hour of actual time. The last two terms are time corrections necessary to convert solar time to LST. The first of these corrections accounts for the difference between LST and solar time on a standard meridian. This correction takes into account the variable speed of the Earth in its path about the sun and the fact that the sun's apparent path lies on the ecliptic rather than on the equatorial plane [4]. This so-called 'Equation of Time' correction is less than 17 minutes on any given date and can be calculated using the formula:

$$C = - 0.12193 \sin B + 0.00839 \cos B - 0.15699 \sin 2B - 0.05391 \cos 2B - 0.00657 \sin 3B - 0.00154 \cos 3B - 0.00370 \sin 4B - 0.00222 \cos 4B \quad (12)$$

where B is given by:

$$B = \frac{(D-0.4)\ 360}{365} \quad (13)$$

The equation of time varies slightly from year to year. Eqn. (12) was determined by fitting Waugh's [4] average daily values of the equation of time with a Fourier series. Waugh states that his average values are accurate to 10 or 15 seconds in any individual year, and Eqn. (12) fits his data to within 10 seconds. Further accuracy is unwarranted since the effect of leap year is not explicitly handled in any of the equations, the effect of the finite disk of the sun is ignored, and local standard times in the accompanying FORTRAN subroutine are printed out with a resolution of only one minute.

The final time correction term in Eqn. (11) accounts for sites that are at longitudes (λ) east or west of the standard meridian (λ_0) of their time zone. For example, a site 1^o of longitude east of a standard meridian will experience sunrise and sunset 4 minutes earlier than a site located on the standard meridian.

3. FORTRAN PROGRAM

A FORTRAN subroutine is presented in Appendix A for calculating instantaneous or daily total extraterrestrial solar flux for a plane surface of arbitrary inclination and azimuth located at a given latitude and longitude on the Earth. Instantaneous fluxes can be calculated for any time of day (LST). Table I below lists the model inputs and outputs. The subroutine would be called from a main program in which the input parameters were defined and the model outputs were to be utilized or listed for the user.

Table I. PROGRAM INPUTS/OUTPUTS

Input/Output	Variable Name	Units	Range
INPUTS:			
Site Parameters:			
Longitude (λ)	LONG	degrees	-180 (W) to 180 (E)
Latitude (ϕ)	LAT	degrees	-90 to 90
Slope Azimuth (a)	AZ	degrees	0 to 360
Slope Inclination (i)	IN	degrees	0 to 90
Time Parameters:			
Month	MO		01 to 12
Day	IDA		01 to 31
Hour	IHR	hours	00 to 23
Minute	MM	minutes	00 to 59
OUTPUTS:			
Option 1: DAILY=.TRUE.			
Daily Total Radiation	OUT1	MJ/m^2	
Sunrise Time	OUT2	hours LST	0 to 24
Sunset Time	OUT3	hours LST	0 to 24
Option 2: DAILY=.FALSE.			
Instantaneous Radiation	OUT1	W/m^2	
Solar Azimuth (A)	OUT2	degrees	0 to 360
Solar Zenith Angle (Z)	OUT3	degrees	0 to 90

The subroutine returns an instantaneous radiation value (OUT1) when the user enters the subroutine with all arguments defined and logical variable DAILY set to (.FALSE.). The sun's azimuth (OUT2) and zenith angle (OUT3) are also returned. A daily total radiation value is returned (OUT1) when the user enters the subroutine with DAILY set to (.TRUE.). In this case the values specified for the time of day (IHR and MM) are not utilized in the subroutine. The sunrise and sunset times (OUT2 and OUT3, respectively) for the inclined surface are also returned.

The course of instantaneous radiation as it changes throughout the day can be easily calculated by constructing a loop in the main program in which the subroutine is called for different times of the day. The subroutine could be called initially with DAILY set to (.TRUE.) to determine the local standard time of sunrise and sunset.

4. EXAMPLE SIMULATION

4.1 Instantaneous Extraterrestrial Radiation

A simple main program can be constructed to call subroutine SOLAR repeatedly at, for example, one-minute intervals of a day to determine how instantaneous radiation would change throughout the course of a day. Figure 1 illustrates the results of such a simulation made at a number of different latitudes on December 22, the northern hemisphere winter solstice. The figure represents the course of instantaneous radiation on any standard meridian at the latitude indicated. Standard meridians are at longitudes of $\pm 15^o \cdot n$, where n is an integer from 0 to 12. Sites not on a standard meridian will have the same course of radiation but, in terms of clock time, will experience sunrise and sunset 4 minutes earlier for every degree of longitude east of the standard meridian. At the North Pole ($\phi = 90^o$) the sun does not rise during the polar night, so that daily total solar flux is zero. At 60^oN a short day is experienced, with the sun rising at 0913 LST and setting at 1444 LST. The short period of daylight provides daily total radiation of only 2.1 MJ/m^2. Days become longer as one travels farther south, and daily totals increase. Daily totals of 19.5, 35.3, and 43.5 MJ/m^2 are experienced at



Figure 1. Instantaneous extraterrestrial solar flux as a function of time on December 22, for latitudes 60^oN, 30^oN, 0^o, 30^oS, 60^oS, and 90^oS.

30°N, at the equator, and at 30°S, respectively. The daylight period continues to increase as the South Pole is approached, but peak instantaneous radiation decreases from nearly 1390 W/m^2 at 30°S to 1123 at 60°S and to 556 at the South Pole [note that the peak value of 1390 W/m^2 at 30°S is slightly larger than the solar constant since the Earth's orbit brings it closest to the sun in the southern hemisphere winter and it receives correspondingly higher radiative fluxes, as given by Eqn. (2)]. Instantaneous solar flux at the South Pole is constant at 556 W/m^2 throughout the entire polar day as the sun circles overhead at a zenith angle of 66.5 degrees.

4.2 Daily Total Extraterrestrial Radiation

Subroutine SOLAR can be called repeatedly from a main program for different days of the year to find the variation in daily totals through the course of a year on a given plane surface at a given location on the Earth. This can be done for plane surfaces of any inclination and azimuth. Figure 2 illustrates this use of the model at 40°N latitude on any standard meridian for surfaces facing south at inclination angles of 0°, 10°, 30°, 50°, 70°, and 90°. A comparison of daily totals on south-facing surfaces of various inclination angles shows that a horizontal surface receives the highest daily total radiation in summer, but receives the lowest total in winter. A south-facing vertical wall (i = 90°) receives only 6.9 MJ/m^2 at the summer solstice, while a plane surface inclined at 70° receives a near-constant 38 MJ/m^2 daily total during a three-month period in midwinter.



Figure 2. Daily totals of extraterrestrial solar flux on a south-facing plane surface during the course of a year. Totals are shown for plane surfaces of inclination, 0°, 10°, 30°, 50°, 70° and 90°.

5. SUMMARY

In this paper we have presented a set of equations that can be used to determine instantaneous and daily total extraterrestrial solar fluxes on plane surfaces of any inclination and azimuth at any latitude and longitude on the Earth. Equations are also given to calculate the sun's position in the sky at any local standard time, and a method has been outlined to determine the local standard times of sunrise and sunset on horizontal and arbitrarily inclined surfaces. A FORTRAN subroutine was presented to apply the method to calculations with digital computers. Example simulations were presented to illustrate the use of the subroutine.

A listing of a FORTRAN program containing the SOLAR subroutine and an interactive main program is available from Environmental Software. The main program is internally documented and is capable of performing calculations in which the various inputs to the subroutine can be incremented automatically within preset limits. The program was developed for a Digital Equipment Corporation VAX 11/780 computer but, with minor modifications, could be adapted to run on other computer systems.

ACKNOWLEDGMENTS

This work was supported by the U.S. Department of Energy at Pacific Northwest Laboratory (PNL) under Contract DE-AC06-76RLO 1830. PNL is operated by Battelle Memorial Institute for the U.S. Department of Energy. Mr. Harlan Foote of PNL provided the Fourier series representing the equation of time.

REFERENCES

[1] Sellers W.D. Physical Climatology, University of Chicago Press, 1965.

[2] Garnier B.J. and Ohmura A. A Method of Calculating the Direct Shortwave Radiation Income of Slopes, J. Appl. Meteor. 1968, 7, 796-800.

[3] McCullough E.C. Total Daily Radiant Energy Available Extraterrestrially as a Harmonic Series in the Day of the Year, Arch. Meteor. Geophys. Biokl., Ser. B, 1968, 16, 129-143.

[4] Waugh A.E. Sundials, Their Theory and Construction, Dover Publications, Inc., 1973.

[5] Kreith F. and Kreider J.F. Principles of Solar Engineering, Hemisphere Publishing Corp., London, 1978.

Appendix A. Listing of FORTRAN Program

```
      SUBROUTINE SOLAR (LONG, LAT, AZ, IN, SC, DAILY, MO, IDA,
     +                  IHR, MM, OUT1, OUT2, OUT3)

C+++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
C     C. D. WHITEMAN AND K. J. ALLWINE                                 +
C     PACIFIC NORTHWEST LABORATORY,  RICHLAND, WA  99352              +
C----------------------------------------------------------------------+
C     USES GARNIER & OHMURA (1968) AND MCCULLOUGH (1968) SCHEMES TO   +
C     CALCULATE EXTRATERRESTRIAL SOLAR RADIATION AT A REQUESTED TIME  +
C     OF DAY ON A SLOPE OF ANY AZIMUTH AND INCLINATION ANGLE GIVEN THE+
C     LATITUDE AND LONGITUDE, DATE, AND SLOPE AZIMUTH AND INCLINATION +
C     ANGLES.  ALSO INTEGRATES INSTANTANEOUS VALUES TO DETERMINE      +
C     THE DAILY TOTAL IF REQUESTED.                                   +
C----------------------------------------------------------------------+
C     INPUTS: LONG = (-180. TO 180. DEG), LAT = (-90. TO 90. DEG),    +
C     AZ = SLOPE AZIMUTH ANGLE (0. TO 359.), IN = SLOPE INCLINATION   +
C     ANGLE (0. TO 90. DEG), SC = SOLAR CONSTANT (E.G. 1353. W/M**2), +
C     DAILY = DAILY TOTAL OR INSTANTANEOUS (.TRUE. OR .FALSE.),       +
C     MO = MONTH (1 TO 12), IDA = DAY (1 TO 31), IHR = HOUR (0 TO 23),+
C     MM = MINUTE (00 TO 59).                                         +
C----------------------------------------------------------------------+
C     OUTPUTS: IF DAILY IS TRUE >> OUT1 = TOTAL RADIATION (MJ/M**2),  +
C     OUT2 = SUNRISE (HOURS LST), OUT3 = SUNSET (HOURS LST).          +
C     IF DAILY IS FALSE >> OUT1 = INSTANTANEOUS RADIATION (W/M**2),   +
C     OUT2 = SUN'S ZENITH ANGL (DEG), OUT3 = SUN'S AZIMUTH ANGL (DEG).+
C+++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++

      LOGICAL DAILY, FIRST
      INTEGER NDAY(12)
      REAL    LAT, LONG, LONGCOR, LONGSUN, IN, INSLO
      REAL    ACOF(4), BCOF(4)

C *** CONSTANTS.
      DATA ACOF/ 0.00839,-0.05391,-0.00154,-0.00222/
      DATA BCOF/-0.12193,-0.15699,-0.00657,-0.00370/
      DATA NDAY/0,31,59,90,120,151,181,212,243,273,304,334/
      DATA DZERO,ECCENT,PI,CALINT/80.,.0167,3.14159,1./
      RTOD = PI/180.
      DECMAX=(23.+26./60.)*RTOD
      OMEGA=2.*PI/365.
      ONEHR=15.*RTOD

C *** JULIAN DATE.
      D=FLOAT(IDA+NDAY(MO))
C *** RATIO OF RADIUS VECTORS SQUARED.
      OMD=OMEGA*D
      OMDZERO=OMEGA*DZERO
      RDVECSQ=1./(1.-ECCENT*COS(OMD))**2
C *** DECLINATION OF SUN.
      LONGSUN=OMEGA*(D-DZERO)+2.*ECCENT*(SIN(OMD)-SIN(OMDZERO))
      DECLIN=ASIN(SIN(DECMAX)*SIN(LONGSUN))
      SDEC=SIN(DECLIN)
      CDEC=COS(DECLIN)
C *** CHECK FOR POLAR NIGHT OR DAY.
      ARG=((PI/2.)-ABS(DECLIN))/RTOD
      IF(ABS(LAT).GT.ARG) THEN
        IF((LAT.GT.0..AND.DECLIN.LT.0.) .OR.
     +     (LAT.LT.0..AND.DECLIN.GT.0.)) THEN
          OUT1=0.
          OUT2=0.
          OUT3=0.
          RETURN
        ENDIF
        SR=-PI
      ELSE
C *** SUNRISE HOUR ANGLE.
        SR=-ABS(ACOS(-TAN(LAT*RTOD)*TAN(DECLIN)))
      ENDIF
C *** STANDARD TIME MERIDIAN FOR SITE.
      STDMRDN=NINT(LONG/15.)*15.
```

```
      LONGCOR=(LONG-STDMRDN)/15.
C *** COMPUTE TIME CORRECTION FROM EQUATION OF TIME.
      B=2.*PI*(D-.4)/365.
      EM=0.
      DO I=1,4
        EM=EM+(BCOF(I)*SIN(I*B)+ACOF(I)*COS(I*B))
      ENDDO
C *** TIME OF SOLAR NOON.
      TIMNOON=12.-EM-LONGCOR

      AZSLO=AZ*RTOD
      INSLO=IN*RTOD
      SLAT=SIN(LAT*RTOD)
      CLAT=COS(LAT*RTOD)
      CAZ=COS(AZSLO)
      SAZ=SIN(AZSLO)
      SINC=SIN(INSLO)
      CINC=COS(INSLO)

      IF (DAILY) THEN
C *** COMPUTE DAILY TOTAL.
        IHR=0
        MM=0
        HINC=CALINT*ONEHR/60.
        IK=(2.*ABS(SR)/HINC)+2.
        FIRST=.TRUE.
        OUT1=0.
        DO I=1,IK
          H=SR+HINC*FLOAT(I-1)
          COSZ=SLAT*SDEC+CLAT*CDEC*COS(H)
          COSBETA=CDEC*((SLAT*COS(H))*(-CAZ*SINC)-
     +            SIN(H)*(SAZ*SINC)+(CLAT*COS(H))*CINC)+
     +            SDEC*(CLAT*(CAZ*SINC)+SLAT*CINC)
          EXTRA=SC*RDVECSQ*COSZ
          IF(EXTRA.LT.0.) EXTRA=0.
          EXTSLO=SC*RDVECSQ*COSBETA
          IF(EXTRA.LE.0. .OR. EXTSLO.LT.0.) EXTSLO=0.
          IF(FIRST .AND. EXTSLO.GT.0.) THEN
            OUT2=(H-HINC)/ONEHR+TIMNOON
            FIRST=.FALSE.
          ENDIF
          IF(.NOT.FIRST .AND. EXTSLO.LE.0.) OUT3=H/ONEHR+TIMNOON
          OUT1=EXTSLO+OUT1
        ENDDO
        OUT1=OUT1*CALINT*60./1000000.
      ELSE
C *** COMPUTE AT ONE TIME.
        T1=FLOAT(IHR)+FLOAT(MM)/60.
        H=ONEHR*(T1-TIMNOON)
        COSZ=SLAT*SDEC+CLAT*CDEC*COS(H)
        COSBETA=CDEC*((SLAT*COS(H))*(-CAZ*SINC)-
     +          SIN(H)*(SAZ*SINC)+(CLAT*COS(H))*CINC)+
     +          SDEC*(CLAT*(CAZ*SINC)+SLAT*CINC)
        EXTRA=SC*RDVECSQ*COSZ
        IF(EXTRA.LT.0.) EXTRA=0.
        EXTSLO=SC*RDVECSQ*COSBETA
        IF(EXTRA.LE.0. .OR. EXTSLO.LT.0.) EXTSLO=0.
        OUT1=EXTSLO
        Z=ACOS(COSZ)
        COSA=(SLAT*COSZ-SDEC)/(CLAT*SIN(Z))
        IF(COSA.LT.-1.) COSA=-1.
        IF(COSA.GT.1.) COSA=1.
        A=ABS(ACOS(COSA))
        IF(H.LT.0.) A=-A
        OUT3=Z/RTOD
        OUT2=A/RTOD+180.
      ENDIF

      RETURN
      END
```